Mini Eliza — 1

```
define myeliza();
    vars x, y;  lvars z;
    [good day - what is your problem] =>
    readline() -> z;
    until z matches [??x bye ??y] do
        respondto(z);
        readline() -> z;
    enduntil;
    [good bye] =>
enddefine;
```

Mini Eliza — 2

```
define respondto(list);
    vars x, y;
    if list matches [??x mother ??y] then
        [tell me more about your family] =>
    elseif list matches [i want to ??x] then
        [do you know anyone else who wants to ^^x] =>
    elseif list matches [i ??x you] then
        [perhaps in your fantasy we ^^x each other] =>
    elseif list matches [??x ill ??y] then
        [have you tried the health centre] =>
    else [how interesting - do go on] =>
    endif;
enddefine;
```

Mini Eliza — 3

```
: myeliza();
** [good day - what is your problem]
? i feel a bit off colour today
** [how interesting - do go on]
? in fact quite ill
** [have you tried the health centre]
? no my mother said it was not sensible
** [tell me more about your family]
? i want to tell you but im shy
** [do you know anyone else who wants to tell you but im shy]
? bye
** [good bye]
```

Mini Parser — 1

```
define isadet(word) -> result;
   member(word, [a the]) -> result
enddefine;

define isanoun(word) -> result;
   member(word, [cat mat milk blanket chair]) -> result
enddefine;

define isaname(word) -> result;
    member(word, [fred mary jack jill]) -> result
enddefine;
define isaverb(word) -> result;
   member(word, [hates teaches studies loves]) -> result
enddefine;
```

Mini Parser — 2

```
define isasent(list) -> result;
    vars x, y, z;
    list matches [??x:isanp ?y:isaverb ??z:isanp] -> result
enddefine;

define isanp(list) -> result;
    vars x, y;
    (list matches [?x:isaname]) or
    (list matches [?x:isadet ?y:isanoun])
    -> result
enddefine;
```

Mini Parser — 3

```
: member("fred", [jack gary larry susan fred jane]) =>
** <true>
: isadet("the") =>
** <true>
: isadet("some") =>
** <false>
: isanp([the chair]) =>
** <true>
: isanp("mary") =>
** <false>
: isanp([mary]) =>
** <true>
: isasent([the cat likes a blanket]) =>
** <false>
: isasent([jack loves a cat]) =>
** <true>
```